CHAPTER 17 A SIMPLE CHEMICAL ENGINEERING FLOWSHEETING EXAMPLE

In this example we shall examine a model for a simple chemical engineering process flowsheet. The code listed below exists in the file in the ASCEND examples subdirectory entitled *simple_fs.asc*. Except for some formatting changes to make it more presentable here, it is exactly as it is in the library version. Thus you could run this example by loading this file and using it and its corresponding script *simple_fs.s*.

17.1 The problem description

This model is of a simple chemical engineering flowsheet. Studying it will help to see how one constructs more complex models in ASCEND. Models for more complex objects are typically built out of previously defined types each of which may itself be built of previously defined parts, etc. A flowsheet could, for example, be built of units and streams. A distillation column could itself be built out of trays and interconnecting streams.

Lines 40 to 56 in the code below give a diagram of the flowsheet we would like to model. This flowsheet is to convert species B into species C. B undergoes the reaction.

B - -> C

The available feed contains 5 mole percent of species A, a light contaminant that acts as an inert in the reactor. We pass this feed into the reactor where only about 7% of B converts per pass. Species C is much less volatile than B which is itself somewhat less volatile than A. Relative volatilities are 12, 10 and 1 respectively for A, B and C. Species A will build up if we do not let it escape from the system. We propose to do this by bleeding off a small portion (say 1 to 2%) of the B we recover and recycle back to the reactor.

The flowsheet contains a mixer where we mix the recycle with the feed, a reactor, a flash unit, and a stream splitter where we split off and remove some of the recycled species B contaminated with species A Our goal is to determine the impact of the bleed on the performance of this flowsheet. We would also like to see if we can run the flash unit to get us fairly pure C as a bottom product from it.

The first type definitions we need for our simple flowsheet are for the variables we would like to use in our model. The ones needed for this example are all in the file atoms.a4l. Thus we will need to load atoms.a4l before we load the file containing the code for this model.

The following is the code for this model. We shall intersperse comments on the code within it.

17.2 THE CODE

As the code is in our ASCEND models directory, it has header information that we require of all such files included as one large comment extending over several lines. Comments are in the form (* comment *).

To assure that appropriate library files are loaded first, ASCEND has the REQUIRE statement, such as appears on line 61:

REQUIRE atoms.a41

This statement causes the system to load the file *atoms.a4l* before continuing with the loading of this file. *atoms.a4l* in turn has a require statement at its beginning to cause *system.a4l* to be loaded before it is.

(**************************************	1
simple_fs.asc	2
by Arthur W. Westerberg	3
Part of the Ascend Library	4
	5
This file is part of the Ascend modeling library.	б
	7
Copyright (C) 1994	8
	9
The Ascend modeling library is free software; you can redistribute	10
it and/or modify it under the terms of the GNU General Public License as	11
published by the Free Software Foundation; either version 2 of the	12
License, or (at your option) any later version.	13
	14
The Ascend Language Interpreter is distributed in hope that it will be	15
useful, but WITHOUT ANY WARRANTY; without even the implied warranty of	16
MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the GNU	17

General Public License for more detail	ls.	1
		1
You should have received a copy of the	e GNU General Public License along	2
with the program; if not, write to the	e Free Software Foundation, Inc.,	2
675 Mass Ave, Cambridge, MA 02139 USA	. Check the file named COPYING.	2
		2
Use of this module is demonstrated by	the associated script file	2
simple_fs.s.		2
*************************************	***************************************	2
		2
(**************************************	*************	2
\$Date: 97/02/20 18:54:21 \$		2
\$Revision: 1.5 \$		3
\$Author: mthomas \$		3
\$Source: /afs/cs.cmu.edu/project/as	cend/Repository/models/examples/	
simple_fs.asc,v \$		3
/******	* * * * * * * * * * * * * * * * * * * *	3
(*		3
		3
The following example illustrates equa	ation based modeling using the	3
ASCEND system. The process is a simp	le recycle process.	3
		3
		3
		4
		4
		4
	split > purge	4
		4
	· · ·	4
	*	4
1		4
v 	I	т 4
		т Л
	 > flach	
		5
		5
		5
		5
		5
	> C	5
		5
This model requires: "system.a41"		5
"atoms.a41"		5
*)		5
		б
REQUIRE atoms.a41		б

The first model we shall define is for defining a stream. In the document entitled "Equation-based Process Modeling" we argue the need to define a stream by maximizing the use of intensive variables and the equations interrelating them. Our problem here requires only the molar flows for the components as the problem definition provides us with all the physical properties as constants. Nowhere for this simple model do we seem to need temperatures, fugacities, etc. To maximize the use of intensive variables, we will use mole fractions and total molar flow to characterize a stream. We must include the equation that says the mole fractions add to unity. Our first model we call *mixture*.

(*	* * * * * * * * * * * * * * * * * * * *	******	62
			63
MO	DEL mixture;		64
			65
	components	IS_A set OF symbol_constant;	66
	y[components]	IS_A fraction;	67
			68
	<pre>SUM[y[i] i IN components]</pre>	= 1.0;	69
			70
ME'	THODS		71
	METHOD clear;		72
	y[components].fixed := FA	LSE;	73
	END clear;		74
			75
	METHOD specify;		76
	y[components].fixed := TR	UE;	77
	y[CHOICE[components]].fix	ed := FALSE;	78
	END specify;		79
			80
	METHOD reset;		81
	RUN clear;		82
	RUN specify;		83
	END reset;		84
			85
EN	D mixture;		86
			87

Line 66 of the model for mixture defines a set of symbol constants. We will later include in this set one symbol constant giving a name for each of the species in the problem (A, B and C). Line 67 defines one mole fraction variable for each element in the set of components, while line 69 says these mole fractions must add to 1.0.

We add a methods section to our model to handle the flag setting which we shall need when making the problem well-posed -- i.e., as a problem having an equal number of unknowns as equations. We first have a method called clear which resets all the "fixed" flags for all the variables in this model to FALSE. This method puts the problem into a known state (all flags are FALSE). The second method is our selection of variables that we wish to fix if we were to solve the equations corresponding to a mixture model. There is only one equation among all the mole fraction variables so we set all but one of the flags to TRUE. The CHOICE function picks arbitrariliy one of the members of the set *components*. For that element, we reset the fixed flag to FALSE, meaning that this one variable will be computed in terms of the values given to the others.

The reset method is useful as it runs first the clear method to put an instance of a mixture model into a known state with respect to its fixed flags, followed by runing the specify method to set all but one of the fixed flags to TRUE.

These methods are not needed to create our model. To include them is a matter of modeling style, a style we consider to be good practice. The investment into writing these methods now has always been paid back in reducing the time we have needed to debug our final models.

The next model we write is for a stream, a model that will include a part we call *state* which is an instance of the type mixture.

(* *******	***************************************	88
		89
MODEL molar_s	tream;	90
		91
components	IS_A set OF symbol_constant;	92
state	IS_A mixture;	93
Ftot,f[components] IS_A molar_rate;	94
		95
components	, state.components ARE_THE_SAME;	96
		97
FOR i IN c	components CREATE	98
f_def[i]: f[i] = Ftot*state.y[i];	99
END;		100
		101
METHODS		102
		103
METHOD cle	ar;	104
RUN sta	te.clear;	105
Ftot.fi	xed := FALSE;	106

```
f[components].fixed:= FALSE;
                                                                                107
   END clear;
                                                                                108
                                                                                109
  METHOD seqmod;
                                                                                110
      RUN state.specify;
                                                                                111
      state.y[components].fixed:= FALSE;
                                                                                112
   END seqmod;
                                                                                113
                                                                                114
  METHOD specify;
                                                                                115
      RUN seqmod;
                                                                                116
      f[components].fixed:= TRUE;
                                                                                117
   END specify;
                                                                                118
                                                                                119
                                                                                120
   METHOD reset;
      RUN clear;
                                                                                121
      RUN specify;
                                                                                122
   END reset;
                                                                                123
                                                                                124
   METHOD scale;
                                                                                125
                                                                                126
      FOR i IN components DO
         f[i].nominal := f[i] + 0.1{mol/s};
                                                                                127
                                                                                128
      END;
      Ftot.nominal := Ftot + 0.1{mol/s};
                                                                                129
   END scale;
                                                                                130
                                                                                131
END molar_stream;
                                                                                132
                                                                                133
```

We define our stream over a set of components. We next include a part which is of type mixture and call it *state* as mentioned above. We also include a variable entitled *Ftot* which will represent the total molar flowrate for the stream. For convenience -- as they are not needed, we also include the molar flows for each of the species in the stream. We realize that the components defined within the part called *state* and the set of components we just defined for the stream should be the same set. We force the two sets to be the same set with the ARE_THE_SAME operator.

We next write the equations that define the individual molar flows for the components in terms of their corresponding mole fractions and the total flowrate for the stream. Note, the equations that says the mole fractions add to unity in the definition of the state forces the total of the individual flowrates to equal the total flowrate. Thus we do not need to include an equation that says the molar flowrates for the species add up to the total molar flowrate for the stream.

	We again write the methods we need for handling flag leave it to the reader to establish that the specify metho well-posed instance involving the same number of vari computed as equations available to compute them. The there as we may occasionally wish to rescale the nomin flows to reflect the values we are computing for them. variables can lead to numerical difficulties for really lar method is there to reduce the chance we will have poor	setting. We od produces a iables to be e scale method is nal values for our Poor scaling of rge models. This r scaling.
	Note that the nominal values for the remaining variable fractions are unity. It does not need to be recompute almost always a good nominal value for each of them.	es the mole ed as unity is
	Our next model is for the first of our unit operations. E be built of streams and equations that characterize their first models a mixer. It can have any number of feed st which is a molar stream. We require the component se feed streams and the output stream from the unit to be Finally we write a component material balance for eac in the problem, where we sum the flows in each of the flow in the output stream, <i>out</i> .	ach of these will r behavior. The treams, each of et for each of the the same set. h of the species feeds to give the
(* **************	*********	134
, , , , , , , , , , , , , , , , , , ,	,	135
MODEL mixer;		136
		137
n_inputs	<pre>IS_A integer_constant;</pre>	138
feed[1n_inputs]	, out IS_A molar_stream;	139
		140
<pre>feed[1n_inputs]</pre>	.components,	141
out.components	ARE_THE_SAME;	142
		143
FOR i IN out.comp	onents CREATE	144
cmb[i]: out.f[:	i] = SUM[feed[1n_inputs].f[i]];	145
END;		146
		147
METHODS		148
		149
METHOD clear;		150
RUN feed[1n_	inputs].clear;	151
RUN out.clear;		152
END Clear;		153
METHOD comed.		154
FND second.		155
Div Degilour		157
		= 57

METHOD specify;	158
RUN seqmod;	159
RUN feed[1n_inputs].specify;	160
END specify;	161
	162
METHOD reset;	163
RUN clear;	164
RUN specify;	165
END reset;	166
	167
METHOD scale;	168
RUN feed[1n_inputs].scale;	169
RUN out.scale;	170
END scale;	171
	172
END mixer;	173

The *METHOD clear* sets all the fixed flags for the parts of this model to false by running each of their clear methods (i.e., for all the feeds and for the stream out). If this model had introduced any new variables, their fixed flags would have been set to FALSE here.

We will implement the method to make the model well posed into two parts: *seqmod* (stands for "sequential modular" which is the mindset we use to get a unit well-posed) and *specify*. The first we shall use within any unit operation to fix exactly enough fixed flags for a unit such that, if we also make the feed streams to it well-posed, the unit will be wellposed. For a mixer unit, the output stream results simply from mixing the input streams; there are no other variables to set other than those for the feeds. Thus the *seqmod* method is empty. It is here for consistency with the other unit operation models we write next. The *METHOD specify* makes this model well-posed by calling the *seqmod* method and then the *specify* method for each of the feed streams. No other flags need be set to make the model well-posed.

METHOD reset simply runs *clear* followed by *specify*. Running this sequence of method will make the problem well-posed no matter which of the fixed flags for it are set to TRUE before running *reset*. Finally, flowrates can take virtually any value so we can include a *scale* method to scale the flows based on their current values.

The next model is for a very simple 'degree of conversion' reactor. The model defines a turnover rate which is the rate at which the reaction as written proceeds (e.g., in moles/s). For example, here our reaction will be B --> C. A turnover rate of 3.7 moles/s would mean that 3.7 moles/s of B would convert to 3.7 moles/s of C. The vector stoich_coef has one

entry per component. Here there will be three components when we test this model so the coefficients would be 0, -1, 1 for the reaction 0 + (-1) + (+1) + (+1) = 0. Reactants have a negative coefficient, products a positive one. The material balance to compute the flow out for each of the components sums the amount coming in plus that created by the reaction.

(**************************************	1/4
	175
MODEL reactor;	176
	177
feed, out IS_A molar_stream;	178
<pre>feed.components, out.components ARE_THE_SAME;</pre>	179
	180
<pre>turnover IS_A molar_rate;</pre>	181
<pre>stoich_coef[feed.components]IS_Afactor;</pre>	182
	183
FOR i IN feed.components CREATE	184
<pre>out.f[i] = feed.f[i] + stoich_coef[i]*turnover;</pre>	185
END;	186
	187
METHODS	188
	189
METHOD clear;	190
RUN feed.clear;	191
RUN out.clear;	192
<pre>turnover.fixed := FALSE;</pre>	193
<pre>stoich_coef[feed.components].fixed := FALSE;</pre>	194
END clear;	195
	196
METHOD seqmod;	197
<pre>turnover.fixed := TRUE;</pre>	198
<pre>stoich_coef[feed.components].fixed := TRUE;</pre>	199
END seqmod;	200
	201
METHOD specify;	202
RUN seqmod;	203
RUN feed.specify;	204
END specify;	205
	206
METHOD reset;	207
RUN Clear;	208
RUN specity;	209
END reset;	210

	211
METHOD scale;	212
RUN feed.scale;	213
RUN out.scale;	214
<pre>turnover.nominal := turnover.nominal+0.0001 {kg_mole/s};</pre>	215
END scale;	216
	217
END reactor;	218
	219

The *METHOD clear* first directs all the parts of the reactor to run their *clear* methods. Then it sets the fixed flags for all variables introduced in this model to FALSE.

Assume the feed to be known. We introduced one stoichiometric coefficient for each component and a turnover rate. To make the output stream well-posed, we would need to compute the flows for each of the component flows leaving. That suggests the material balances we wrote are all needed to compute these flows. We would, therefore, need to set one fixed flag to TRUE for each of the variables we introduced, which is what we do in the *METHOD seqmod*. Now when we run *seqmod* and then the *specify* method for the feed, we will have made this model well-posed, which is what we do in the *METHOD specify*.

The flash model that follows is a constant relative volatility model. Try reasoning why the methods attached are as they are.

(* *********	*****	* * * * * * * * * * * * * * *)	220
			221
MODEL flash;			222
			223
feed,vap,liq I	.S_A mola	ar_stream;	224
			225
feed.components,			226
vap.components,			227
liq.components	ARE_	_THE_SAME;	228
			229
alpha[feed.componer	its],		230
ave_alpha	IS_A	A factor;	231
			232
<pre>vap_to_feed_ratio</pre>	IS_A	A fraction;	233
			234
<pre>vap_to_feed_ratio*f</pre>	eed.Ftot = vap	.Ftot;	235
			236
FOR i IN feed.compo	onents CREATE		237
cmb[i]: feed.f[i] = vap.f[i] +	liq.f[i];	238

```
eq[i]: vap.state.y[i]*ave_alpha = alpha[i]*liq.state.y[i];
                                                                      239
  END;
                                                                      240
                                                                      241
METHODS
                                                                      242
                                                                      243
  METHOD clear;
                                                                      244
     RUN feed.clear;
                                                                      245
     RUN vap.clear;
                                                                      246
     RUN liq.clear;
                                                                      247
     alpha[feed.components].fixed := FALSE;
                                                                      248
     ave_alpha.fixed
                                   := FALSE;
                                                                      249
     vap_to_feed_ratio.fixed := FALSE;
                                                                      250
  END clear;
                                                                      251
                                                                      252
  METHOD seqmod;
                                                                      253
     alpha[feed.components].fixed := TRUE;
                                                                      254
     vap_to_feed_ratio.fixed := TRUE;
                                                                      255
  END seqmod;
                                                                      256
                                                                      257
  METHOD specify;
                                                                      258
     RUN seqmod;
                                                                      259
     RUN feed.specify;
                                                                      260
  END specify;
                                                                      261
                                                                      262
  METHOD reset;
                                                                      263
     RUN clear;
                                                                      264
                                                                      265
     RUN specify;
  END reset;
                                                                      266
                                                                      267
  METHOD scale;
                                                                      268
     RUN feed.scale;
                                                                      269
     RUN vap.scale;
                                                                      270
     RUN liq.scale;
                                                                      271
  END scale;
                                                                      272
                                                                      273
END flash;
                                                                      274
                                                                      275
276
                                                                      277
```

The final unit operation model is the splitter. The trick here is to make all the states for all the output streams the same as that of the feed. This move makes the compositions all the same and introduces only one equation to add those mole fractions to unity. The rest of the model should be evident.

```
MODEL splitter;
                                                                             278
                                                                             279
  n_outputs
                                IS_A integer_constant;
                                                                             280
   feed, out[1..n_outputs ]
                                IS_A molar_stream;
                                                                             281
                                IS_A fraction;
                                                                             282
   split[1..n_outputs]
                                                                             283
                                                                             284
   feed.components, out[1..n_outputs].components ARE_THE_SAME;
                                                                             285
   feed.state,
                                                                             286
   out[1..n_outputs].state ARE_THE_SAME;
                                                                             287
                                                                             288
   FOR j IN [1...n_outputs] CREATE
                                                                             289
      out[j].Ftot = split[j]*feed.Ftot;
                                                                             290
                                                                             291
   END;
                                                                             292
   SUM[split[1..n_outputs]] = 1.0;
                                                                             293
                                                                             294
METHODS
                                                                             295
                                                                             296
  METHOD clear;
                                                                             297
      RUN feed.clear;
                                                                             298
      RUN out[1..n outputs].clear;
                                                                             299
      split[1..n_outputs-1].fixed:=FALSE;
                                                                             300
   END clear;
                                                                             301
                                                                             302
  METHOD seqmod;
                                                                             303
                                                                             304
      split[1..n_outputs-1].fixed:=TRUE;
   END seqmod;
                                                                             305
                                                                             306
  METHOD specify;
                                                                             307
      RUN seqmod;
                                                                             308
      RUN feed.specify;
                                                                             309
   END specify;
                                                                             310
                                                                             311
  METHOD reset;
                                                                             312
      RUN clear;
                                                                             313
                                                                             314
      RUN specify;
   END reset;
                                                                             315
                                                                             316
  METHOD scale;
                                                                             317
      RUN feed.scale;
                                                                             318
      RUN out[1..n_outputs].scale;
                                                                             319
   END scale;
                                                                             320
                                                                             321
END splitter;
                                                                             322
                                                                             323
```

(*	***************************************	*)	324
(^	· · · · · · · · · · · · · · · · · · ·	^)	34

324 325

Now we shall see the value of writing all those methods for our unit operations (and for the models that we used in creating them). We construct our flowsheet by saying it includes a mixer, a reactor, a flash unit and a splitter. The mixer will have two inputs and the splitter two outputs. The next few statements configure our flowsheet by making, for example, the output stream from the mixer and the feed stream to the reactor be the same stream.

The methods are as simple as they look. This model does not introduce any variables nor any equations that are not introduced by its parts. We simply ask the parts to clear their variable fixed flags.

To make the flowsheet well-posed, we ask each unit to set sufficient fixed flags to TRUE to make itself well posed were its feed stream wellposed (now you can see why we wanted to create the methods *seqmod* for each of the unit types.) Then the only streams we need to make well-posed are the feeds to the flowsheet, of which there is only one. The remaining streams come out of a unit which we can think of computing the flows for it.

MODEL flowsheet;		326
		327
ml	IS_A mixer;	328
rl	IS_A reactor;	329
fll	IS_A flash;	330
spl	IS_A splitter;	331
		332
(* define sets *)		333
		334
ml.n_inputs	:==2;	335
spl.n_outputs	:==2;	336
		337
(* wire up flowsheet *)		338
		339
ml.out, rl.feed	ARE_THE_SAME;	340
rl.out, fll.feed	ARE_THE_SAME;	341
fll.vap, spl.feed	ARE_THE_SAME;	342
<pre>spl.out[2], ml.feed[2]</pre>	ARE_THE_SAME;	343
		344
METHODS		345
		346
METHOD clear;		347
RUN m1.clear;		348

RUN r1.clear;	349
RUN fll.clear;	350
RUN spl.clear;	351
END clear;	352
	353
METHOD seqmod;	354
RUN ml.seqmod;	355
RUN r1.seqmod;	356
RUN fll.seqmod;	357
RUN spl.seqmod;	358
END seqmod;	359
	360
METHOD specify;	361
RUN seqmod;	362
RUN m1.feed[1].specify;	363
END specify;	364
	365
METHOD reset;	366
RUN clear;	367
RUN specify;	368
END reset;	369
	370
METHOD scale;	371
RUN m1.scale;	372
RUN r1.scale;	373
RUN fl1.scale;	374
RUN spl.scale;	375
END scale;	376
	377
END flowsheet;	378
	379
(* ************************************	380
	381

We have created a flowsheet model above. If you look at the reactor model, we require that you specify the turnover rate for the reaction. We may have no idea of a suitable turnover rate. What we may have an idea about is the conversion of species B in the reactor; for example, we may know that about 7% of the B entering the reactor may convert. How can we alter our model to allow for us to say this about the reactor and not be required to specify the turnover rate? In a sequential modular flowsheeting system, we would use a "computational controller." We shall create a model here that gives us this same functionality. Thus we call it a "controller." There are many ways to construct this model. We choose here to create a model that has a flowsheet as a part of it. We introduce a variable conv which will indicate the fraction conversion of any one of the components which we call the key_component here. For that component, we add a material balance based on the fraction of it that will convert. We added one new variable and one new equation so, if the flowsheet is wellposed, so will our controller be well-posed. However, we want to specify the conversion rather that the turnover rate. The *specify* method first asks the flowsheet fs to make itself well-posed. Then it makes this one trade: fixing conv and releasing the turnover rate.

MODEL controller;			382
			383
fs	IS_A	flowsheet;	384
conv	IS_A	fraction;	385
key_components	IS_A	<pre>symbol_constant;</pre>	386
<pre>fs.r1.out.f[key_components] =</pre>	(1 - con	<pre>w)*fs.rl.feed.f[key_components];</pre>	387
			388
METHODS			389
			390
METHOD clear;			391
RUN fs.clear;			392
<pre>conv.fixed:=FALSE;</pre>			393
END clear;			394
			395
METHOD specify;			396
RUN fs.specify;			397
fs.r1.turnover.fixed:=FALSE	;		398
<pre>conv.fixed:=TRUE;</pre>			399
END specify;			400
			401
METHOD reset;			402
RUN clear;			403
RUN specify;			404
END reset;			405
			406
METHOD scale;			407
RUN fs.scale;			408
END scale;			409
			410
END controller;			411
			412
(* ********	* * * * * * * *	************	413
			414

We now would like to test our models to see if they work. How can we write test for them? We can create test models as we do below.

To test the flowsheet model, we create a test_flowsheet model that refines our previously defined flowsheet model. "To refine the previous model" means this model includes all the statements made to define the flowsheet model plus those statements that we now provide here. So this model is a flowsheet but with it components specified to be 'A', 'B', and 'C'. We add a new method called *values* in which we specify values for all the variables we intend to fix when we solve. We can also provide values for other variables; these will be used as the initial values for them when we start to solve. We see all the variables being given values with the units specified. The units must be specified in ASCEND. ASCEND will interpret the lack of units to mean the variable is unitless. If it is not, then you will get a diagnostic from ASCEND telling you that you have written a dimensionally inconsistent relationship.

Note we specify the molar flows for the three species in the feed. Given these flows, the equations for the stream will compute the total flow and then the mole fractions for it. Thus the feed stream is fully specified with these flows.

We look at the seqmod method for each of the units to see the variables to which we need to give values here.

```
MODEL test flowsheet REFINES flowsheet;
                                                                             415
                                                                             416
  m1.out.components:==['A','B','C'];
                                                                             417
                                                                             418
  METHODS
                                                                            419
                                                                            420
  METHOD values;
                                                                            421
      ml.feed[1].f[`A']
                                 := 0.005 {kg_mole/s};
                                                                            422
      m1.feed[1].f[`B']
                                 := 0.095 {kg_mole/s};
                                                                            423
      m1.feed[1].f[`C']
                                 := 0.0 {kg_mole/s};
                                                                             424
                                                                            425
      r1.stoich coef['A']
                                 := 0;
                                                                            426
      r1.stoich coef['B']
                                 := −1;
                                                                             427
                                 := 1;
      r1.stoich coef['C']
                                                                             428
                                   := 3 {kg_mole/s};
                                                                            429
      r1.turnover
                                                                             430
                                                                            431
      fl1.alpha[`A']
                                 := 12.0;
      fl1.alpha[`B']
                                 := 10.0;
                                                                            432
                                 := 1.0;
      fl1.alpha[`C']
                                                                            433
      fl1.vap_to_feed_ratio
                                := 0.9i
                                                                            434
                                 := 5.0;
                                                                            435
      fl1.ave_alpha
                                                                             436
      spl.split[1]
                                 := 0.01;
                                                                             437
```

	438
<pre>fl1.liq.Ftot:=m1.feed[1].f[`B'];</pre>	439
END values;	440
	441
END test_flowsheet;	442
	443
(* ************************************	444
	445

Finally we would like to test our controller model. Again we write our test model as a refinement of the model to be tested. The test model is, therefore, a controller itself. We make our fs model inside our test model into a test_flowsheet, making it a more refined type of part than it was in the controller model. We can do this because the test_controller model is a refinement of the flowsheet model which fs was previously. A test_flowsheet is, as we said above, a flowsheet. We create a values method which first runs the values method we wrote for the test_flowsheet model and then adds a specification for the conversion of B in the reactor.

MODEL test_controller REFINES controller;		446
		447
fs IS_REFII	NED_TO test_flowsheet;	448
key_components :	==`B';	449
		450
METHODS		451
		452
METHOD values;		453
RUN fs.values	;	454
conv	:= 0.07;	455
END values;		456
		457
END test_controller;		458
		459
(* ************************************		460
		461

160